What is the ratio of a N[0,1] and U[-1/2,1/2]? I have come across a problem where I can reasonably assume that the numerator is a uniform distribution of the type U[-a,a], i.e., centered on zero, and the denominator is N[0,b]. This seems to be Cauchy, or at least that is the best answer from simulations and search of best fitting distribution types. Similarly, if I use $\dfrac{N[0,b]}{U[-a,a]}$ rather than $\dfrac{U[-a,a]}{N[0,b]}$, it still seems to be Cauchy. If I use $\dfrac{U[-a,a]}{U[-b,b]}$, it is close to being Cauchy, and as we well know $\dfrac{N[0,a]}{N[0,b]}$ is Cauchy.
Any insights as to what is going on?
BTW, there was a superficially similar question, which does not seem to use zero centered uniform distributions, but seems irrelevant to my current problem. 
 A: For two independent "benignly" distributed random variables $X$ and $Y$, the ratio $Z = X/Y$ has the pdf
$$p_Z(z) = z^{-2} \int_{-\infty}^\infty |x|\, p_X(x)\, p_Y(x/z)\, dx \sim z^{-2}\, p_Y(0)\, \langle|X|\rangle, \quad |z| \to \infty.$$
This indicates why the tails of the ratio are Cauchy-like given that the denominator $Y$ has a smooth finite pdf at zero.
A: This gives two answers but not the algebraic steps to prove it.  Using Mathematica one can find the density for the ratio $N/U$:
d = TransformedDistribution[z/u, {z \[Distributed] NormalDistribution[0, b], 
    u \[Distributed] UniformDistribution[{-a, a}]}];
PDF[d, r]

$$\frac{b \left(1-e^{-\frac{a^2 r^2}{2 b^2}}\right)}{\sqrt{2 \pi } a r^2}$$
For the density at $r=0$, one could take the limit of the above function as $r \rightarrow 0$:
Limit[PDF[d, r], r -> 0]

$$\frac{a}{2 \sqrt{2 \pi } b}$$
As a check one could perform some simulations:
n = 10000;
zz = RandomVariate[NormalDistribution[0, 1], n];
uu = RandomVariate[UniformDistribution[{-1/2, 1/2}], n];
rr = zz/uu;]
density = PDF[d /. {a -> 1/2, b -> 1}, r];
Show[Histogram[rr, {0.5}, "PDF", PlotRange -> {{-10, 10}, All}],
 Plot[density, {r, -10, 10}]]


For the ratio $U/N$ one can find the density as follows:
d = TransformedDistribution[u/z, {z \[Distributed] NormalDistribution[0, b], 
    u \[Distributed] UniformDistribution[{-a, a}]}];
PDF[d, r]

$$\begin{array}{cc}
 \{ & 
\begin{array}{cc}
 \frac{b}{\sqrt{2 \pi } a} & r=0 \\
 \frac{b \left(1-e^{-\frac{a^2}{2 b^2 r^2}}\right)}{\sqrt{2 \pi } a} & r \neq 0 \\
\end{array}
 \\
\end{array}$$
Here, too, simulations help:
n = 10000;
zz = RandomVariate[NormalDistribution[0, 1], n];
uu = RandomVariate[UniformDistribution[{-1/2, 1/2}], n];
rr = uu/zz;
density = PDF[d, r] /. {a -> 1/2, b -> 1};
Show[Histogram[rr, {0.1}, "PDF"],
 Plot[density, {r, -5, 5}, PlotRange -> {{-4, 4}, All}], 
 PlotRange -> {{-4, 4}, All}] 


